

```

2 TRAP 6998:GOSUB 6504:CLOSE #1:POKE 731,1
3 DIM DSC$(100):DIM MAPX(1):DIM MAPY(1):DIM MAP$(252):GOSUB 6510:MAPX(1)=0
5 DIM LOCDR(6):DIM DESC$(100):DIM LOOKDIR(1):DIM NOTE1$(200):DIM MAPVISIBLE(1):DIM ITEMSININV(1)
7 DIM LOCITEMX(MAXITEMS):DIM LOCITEMY(MAXITEMS):DIM ITEMINLOC(MAXITEMS)
8 DIM PLACE$(200):DIM C$(20):DIM VERB$(20):DIM INV$(20):DIM INV(MAXITEMS):DIM LOCX(1):DIM LOCY(1):DIM TMP$(20)
9 DIM LOCN(1):DIM LOCS(1):DIM LOCE(1):DIM LOCWY(1):DIM OLDX(1):DIM OLDY(1):DIM TMP1$(20)
12 DIM MAPNR(1):DIM MAPNRX(2):DIM MAPNRY(2)
49 GRAPHICS 0:POKE 559,0:GOSUB 9002
50 CLOSE #1:OPEN #1,4,0,"D:\WELCOME.TXT":FOR I=1 TO 24:INPUT #1,PLACE$:? PLACE$:NEXT I:POKE 559,34
51 ? :? :? "Co robisz ";:INPUT VERB$
52 IF VERB$="GRAJ" THEN GOTO 80
53 IF VERB$="KONIEC" THEN END :REM KONIEC
72 RUN
80 REM POCZATEK GRY
85 AP=130
89 MAPNRX(1)=MAPX:MAPNRY(1)=5:MAPNRX(2)=MAPX:MAPNRY(2)=13:MAPNRY(0)=1:MAPNRX(0)=MAPX:LOCX=6:LOCY=3:C$="CO
ROBISZ "
90 FOR I=1 TO MAXITEMS:INV(I)=0:ITEMINLOC(I)=0:NEXT I
92 GOSUB 6508
93 LOCITEMX(9)=3:LOCITEMY(9)=9:ITEMINLOC(9)=1:ITEMINLOC(12)=2
94 ITEMINLOC(5)=1:LOCITEMX(5)=2:LOCITEMY(5)=14:ITEMINLOC(3)=1:LOCITEMX(3)=8:LOCITEMY(3)=3
95 ITEMINLOC(1)=1:LOCITEMX(1)=9:LOCITEMY(1)=2:ITEMINLOC(6)=1:LOCITEMX(6)=5:LOCITEMY(6)=15:LOOKDIR=1:VIEWON=0
96 ITEMINLOC(7)=1:LOCITEMX(7)=3:LOCITEMY(7)=6:ITEMINLOC(8)=1:LOCITEMX(8)=6:LOCITEMY(8)=2:MAPVISIBLE=0:INCAVE=0
97 ITEMHERE=0:INV(2)=1:INV(4)=1:INV(11)=1:ITEMSININV=4:GRAPHICS 7:POKE 559,0:GOSUB 9002:DL=PEEK(560)+256*PEEK(561)
98 POKE DL+5,AP:POKE DL+4,0:MAPNR=1:? "Wchodzisz do recepcji fortu. Slyszysz jakis szum... ";:POKE 559,34
99 ? "Kreci ci sie w glowie...":GOSUB 900:C=1:GOSUB 840
100 PRINT :PRINT C$;:INPUT VERB$:PRINT
105 GOSUB 415:GOSUB 421:GOSUB 900
111 IF VERB$="LOOKLOC" THEN GOSUB 9000
112 IF VERB$="MAP" THEN GOSUB 798
113 IF VERB$="LOOK" THEN GOSUB 6452
114 IF VERB$="OPEN" THEN GOSUB 204
115 IF VERB$="INV" THEN GOSUB 906
116 IF VERB$="EXIT" THEN PRINT :PRINT "KONIEC GRY. DZIEKUJEMY. ";:GOSUB 810:GOTO 6500
117 IF LOCX=2 AND LOCY=4 THEN GOSUB 231
118 IF VERB$="DROP" THEN GOSUB 6446
119 IF VERB$="ENTER" THEN GOSUB 380:GOSUB 220
120 IF VERB$="USE" THEN GOSUB 300
121 ITEMHERE=0:IF VERB$="GET" THEN GOSUB 6440
123 IF VERB$="EXAMINE" THEN GOSUB 500
124 GOSUB 380:GOSUB 383
125 IF VERB$="VIEW" AND VIEWON=0 THEN VIEWON=1:GOSUB 380:POKE 89,144:POKE 88,96:CH=1:GOSUB 9010:GOTO 127
126 IF VERB$="VIEW" AND VIEWON=1 THEN C=0:GOSUB 820:VIEWON=0:GOSUB 1000:IF MAPVISIBLE=0 THEN POKE DL+5,AP:POKE DL+4,0
127 IF VIEWON=1 AND CH=1 THEN GOSUB 900:C=255:GOSUB 820:GOSUB 1000
128 IF VERB$="SAVE" THEN GOSUB 1200
129 IF VERB$="LOAD" THEN POKE DL+5,AP:POKE DL+4,0:GOSUB 1210
131 IF VERB$="NOTE" THEN GOSUB 795
132 IF VERB$="READ" THEN GOSUB 793
135 CH=0:GOSUB 207:PRINT :GOTO 100
198 Y2=MAPNRY(MAPNR-1):Y=MAPNRY(MAPNR)+Y2-1:IF MAPNR>1 THEN Y=Y+1
199 FOR I=1 TO MAPY:POSITION 121,20+I:? #6;"0000000000000000":NEXT I
200 IF MAPVISIBLE=1 THEN GOTO 202
201 RETURN
202 FOR I=1 TO MAPY:IF I>=Y2 AND I<=Y THEN POSITION 121,20+I:? #6;MAP$((I-1)*MAPX+1,(I-1)*MAPX+MAPX)
203 NEXT I:CH=0:POKE 764,255:RETURN
204 RETURN :REM OPEN COMMAND
207 ITEMHERE=0:FOR I=1 TO MAXITEMS:IF ITEMINLOC(I)=1 AND LOCITEMX(I)=LOCX AND LOCITEMY(I)=LOCY THEN ITEMHERE=1
208 NEXT I:IF ITEMHERE=1 THEN PRINT "Tu cos jest. ";
209 ITEMHERE=0:RETURN
210 OPEN #1,4,0,"H1:LOCS.DSC":INPUT #1,PLACE$:IF LOCX=8 AND LOCX=3 THEN INPUT #1,PLACE$:INPUT #1,PLACE$
211 CLOSE #1
220 OLDX=LOCX:OLDY=LOCY:CH=0

```

```

225 IF LOCX=9 AND LOCY=4 AND MAPNR=1 THEN LOCY=6:PRINT "Wszedles na dolny poziom fortu. ";CH=1:MAPNR=2:GOTO 227
226 IF LOCX=9 AND LOCY=6 AND MAPNR=2 THEN LOCY=4:PRINT "Wszedles do holu na gornym poziomie fortu. ";CH=1:MAPNR=1
227 IF CH=1 THEN GOSUB 870:GOSUB 900
228 GOSUB 700:IF CH=1 AND VIEWON=1 THEN GOSUB 870:C=255:GOSUB 820
229 IF MAPVISIBLE=1 THEN GOSUB 198
230 GOSUB 900:RETURN
231 GOSUB 350
232 RETURN
235 IF (SALA=1) AND (INV(2)=1) AND (LOCX=2) AND (LOCY=10) THEN GOTO 237
236 PRINT :RETURN
237 PRINT "Wydziabales ogromna dziure w scianie. Widac przejscie... Sztylet wypadl ci z reki i znikl w ciemnosci. ";
238 INV(2)=0:SALA=0
239 GOSUB 810:PRINT
240 RETURN
260 IF MAPVISIBLE=1 THEN MAPVISIBLE=0:GOSUB 198:GOSUB 380:GOSUB 265:? "Zamknales mape.":RETURN
262 IF MAPVISIBLE=0 THEN GOSUB 380:MAPVISIBLE=1:? "Aktualizacja mapy. Czekaj...":GOSUB 198
263 POKE DL+5,144:POKE DL+4,96:? "Mapa gotowa. Mozesz grac dalej...":GOSUB 810:? :RETURN
265 IF VIEWON=0 THEN POKE DL+5,AP:POKE DL+4,0
266 RETURN
270 ? "Przedmioty w plecaku: "
271 GOSUB 6504:C=0:FOR K=1 TO MAXITEMS:INPUT #1,INV$:IF ITEMINLOC(K)=2 THEN GOTO 276
272 NEXT K:CLOSE #1:IF C=0 THEN ? "plecak jest pusty."
273 GOSUB 810:RETURN
276 C=1:? INV$;:INPUT VERB$
277 NEXT K
280 L=0:FOR K=1 TO MAXITEMS:IF ITEMINLOC(K)=2 THEN L=1
281 NEXT K:IF L=0 THEN RETURN
282 ? "Ktory przedmiot bierzesz z plecaka";:INPUT VERB$
283 GOSUB 6504:FOR K=1 TO MAXITEMS:INPUT #1,INV$
284 IF INV$=VERB$ AND ITEMINLOC(K)=2 THEN INV(K)=1:ITEMINLOC(K)=0:K=MAXITEMS:? "Wziales ";VERB$;" z ";TMP$;". ";
286 NEXT K:RETURN
300 ? :? :? "Co chcesz uzyc ";:INPUT TMP$:?
301 IF TMP$="PRZYCISK" AND LOCX=2 AND LOCY=4 THEN GOTO 316
309 GOSUB 360
310 REM IF I=2 THEN GOTO 235
311 IF I=8 THEN GOTO 260
312 IF I=3 AND (LOCX=3 OR LOCX=5) AND LOCY=10 THEN DOOR=DOOR+1:GOTO 314
313 RETURN
314 IF DOOR=2 THEN DOOR=0:PRINT "Otworzyles kluczem drzwi do celi.";
315 IF DOOR=1 THEN ? "Zamknales kluczem drzwi do celi.";
316 IF LOCX=2 AND LOCY=4 THEN ? "Uslyszales zgrzyt otwieranych lub zamykanych zamkow gdzies w okolicy.";
317 IF LOCX=2 AND LOCY=4 THEN LOOKDIR(0)=LOOKDIR(0)+1:IF LOOKDIR(0)=2 THEN LOOKDIR(0)=0
318 RETURN
321 PRINT :PRINT "Nie mozesz tego uzyc.";
322 RETURN
330 INV(1)=0:INV(2)=0:INV(10)=1:RETURN
350 IF INV(3)=1 AND INV(5)=1 THEN GOTO 352
351 RETURN
352 IF (LOCITEMX(1)=10 AND LOCITEMY(1)=4) OR (LOCITEMX(10)=10 AND LOCITEMY(10)=4) THEN GOTO 6499
353 IF LOCITEMX(9)=10 AND LOCITEMY(9)=4 AND (ITEMINLOC(10)=2 OR ITEMINLOC(1)=2) THEN GOTO 6499
354 RETURN
360 GOSUB 6504:FOR I=1 TO MAXITEMS:INPUT #1,VERB$:IF TMP$=VERB$ AND INV(I)=1 THEN GOTO 363
361 NEXT I
362 PRINT "Nie mozesz tego uzyc. ";:GOSUB 810:RETURN
363 IF I=1 OR I=2 OR I=9 OR I=13 THEN PRINT :PRINT "Z czym chcesz uzyc ";:PRINT TMP$;:INPUT VERB$:GOTO 365
364 CLOSE #1:RETURN
365 GOSUB 6504:FOR J=1 TO MAXITEMS:INPUT #1,TMP1$:IF TMP1$=VERB$ AND INV(J)=1 THEN GOTO 367
366 NEXT J:PRINT :GOTO 362
367 IF (I=1 AND J=2) OR (I=2 AND J=1) THEN PRINT :PRINT "Masz wrzatek w kubku.":GOTO 379
368 IF (I=9 AND J<>9) OR I=13 THEN INV(J)=0:? "Wlozyles ";TMP1$;" do ";TMP$;". "
369 IF I=9 AND J<>9 THEN ITEMINLOC(J)=2:IF J=8 THEN MAPVISIBLE=1:GOSUB 260
378 RETURN

```

```

379 INV(1)=0:INV(2)=0:INV(10)=1:RETURN
380 A=OLDY:IF A<1 THEN A=MAPY+LOCY:IF OLDY<1 THEN OLDY=OLDY+MAPY
381 IF MAPVISIBLE=1 THEN COLOR 0:PLOT OLDX+120,A+20
382 RETURN
383 A=LOCY:IF A<1 THEN A=MAPY+LOCY
384 IF MAPVISIBLE=1 THEN COLOR 2:PLOT LOCX+120,A+20
385 RETURN
400 L=LOOKDIR:IF VERB$="B" THEN L=LOOKDIR+2:IF L>4 THEN L=L-4
410 IF L=1 THEN VERB$="N"
411 IF L=2 THEN VERB$="E"
412 IF L=3 THEN VERB$="S"
413 IF L=4 THEN VERB$="W"
414 CH=0
415 IF VERB$="S" THEN IF LOCS=0 THEN PRINT "Idziesz na poludnie. ";:GOSUB 870:OLDX=LOCX:OLDY=LOCY:LOCY=LOCY+1:CH=1
416 IF VERB$="E" THEN IF LOCE=0 THEN PRINT "Idziesz na wschod. ";:GOSUB 870:OLDX=LOCX:OLDY=LOCY:LOCX=LOCX+1:CH=1
417 IF VERB$="N" THEN IF LOCN=0 THEN PRINT "Idziesz na polnoc. ";:GOSUB 870:OLDX=LOCX:OLDY=LOCY:LOCY=LOCY-1:CH=1
418 IF VERB$="W" THEN IF LOCW=0 THEN PRINT "Idziesz na zachod. ";:GOSUB 870:OLDX=LOCX:OLDY=LOCY:LOCX=LOCX-1:CH=1
419 IF MAPVISIBLE=1 AND CH=1 THEN GOSUB 380:GOSUB 383
420 RETURN
421 IF VERB$="L" THEN GOSUB 870:CH=1:IF LOOKDIR>0 THEN LOOKDIR=LOOKDIR-1
422 IF VERB$="L" THEN GOSUB 870:CH=1:IF LOOKDIR=0 THEN LOOKDIR=4
423 IF VERB$="R" THEN GOSUB 870:CH=1:IF LOOKDIR<5 THEN LOOKDIR=LOOKDIR+1
424 IF VERB$="R" THEN GOSUB 870:CH=1:IF LOOKDIR=5 THEN LOOKDIR=1
425 IF VERB$="F" THEN GOSUB 400:IF CH=1 THEN GOSUB 870
426 IF VERB$="B" THEN GOSUB 400:IF CH=1 THEN GOSUB 870
427 RETURN
500 CH=0:?:?:?"Co sprawdzasz ";:INPUT TMP$:PRINT :GOSUB 507:IF CH=1 THEN GOTO 506
502 GOSUB 6504:FOR I=1 TO MAXITEMS:INPUT #1,VERB$:IF TMP$=VERB$ AND INV(I)=1 THEN GOTO 504
503 NEXT I:PRINT "Nie posiadasz tego w inwentarzu. ";:GOSUB 810:RETURN
504 GOSUB 6505:FOR J=1 TO I:INPUT #1,DESC$:NEXT J
505 IF I=9 THEN ??:DESC$;:GOSUB 810:?:CH=0:CLOSE #1:GOSUB 270:RETURN
506 ??:DESC$;:GOSUB 810:PRINT :CH=0:CLOSE #1:RETURN
507 GOSUB 6507:FOR I=1 TO K:INPUT #1,C,D,E,L,VERB$
508 IF TMP$=VERB$ AND LOCX>=C AND LOCY>=D AND LOCX<=E AND LOCY<=L THEN CH=1:I=K+1
509 INPUT #1,DESC$:NEXT I:CLOSE #1:RETURN
700 IF LOCX=5 AND LOCY=10 AND DOOR=0 THEN LOCX=3:PRINT "Wszedles do celi nr 10/5. ";:CH=1:RETURN
701 IF LOCX=3 AND LOCY=10 AND DOOR=0 THEN LOCX=5:PRINT "Wszedles na korytarz. ";:CH=1:RETURN
702 IF LOCX=2 AND LOCY=2 AND LOOKDIR(0)=0 THEN GOTO 770
703 IF LOCX=7 AND LOCY=-1 AND LOOKDIR(0)=0 THEN GOTO 774
704 IF LOCX=13 AND LOCY=15 THEN LOCY=13:?"Wszedles do stolowki. ";:CH=1:RETURN
705 IF LOCX=13 AND LOCY=13 THEN LOCY=15:?"Wszedles ze stolowki na korytarz. ";:CH=1:RETURN
706 IF LOCX=13 AND LOCY=10 THEN LOCY=8:?"Wszedles do kuchni. ";:CH=1:RETURN
707 IF LOCX=13 AND LOCY=8 THEN LOCY=10:?"Wszedles do stolowki. ";:CH=1:RETURN
750 RETURN
770 GOSUB 870:LOCX=7:LOCY=-1:MAPNR=2:?"Przeszedles przez masywna brame wglab fortu.";
771 MAPX(1)=1:GOSUB 6510:MAPNRX(MAPNR)=MAPX:MAPNRY(MAPNR)=13
772 CH=1:GOSUB 900
773 RETURN
774 GOSUB 870:LOCX=2:LOCY=2:MAPNR=1:?"Wszedles na obszar gornego poziomu fortu.";
775 MAPX(1)=0:GOSUB 6510:MAPNRX(MAPNR)=MAPX:MAPNRY(MAPNR)=5
776 CH=1:GOSUB 900
777 RETURN
793 IF INV(12)=1 THEN ?:"Notatka ";:?:NOTE1$:GOSUB 810:?:RETURN
794 IF INV(12)=0 THEN ?:"Nie masz przy sobie notatnika. ":RETURN
795 IF INV(11)=1 AND INV(12)=1 THEN ?:"Napisz notatke ";:INPUT NOTE1$:?"Notatka zostala zapisana. ":RETURN
796 IF INV(11)=0 THEN ?:"Nie masz czym zapisac notatki. ":RETURN
797 IF INV(12)=0 THEN ?:"Nie masz przy sobie notatnika. ":RETURN
798 IF INV(8)=1 THEN GOSUB 260:RETURN
799 ?:"Nie posiadasz mapy. ";:RETURN
800 ??:?"Nie posiadasz podgladu lokacji. ";:GOSUB 810:PRINT :?:RETURN
810 POKE 764,255:?" Nacisnij [RETURN]";:INPUT DSC$:RETURN
816 PLOT 62,15:DRAWTO 70,11:DRAWTO 95,11:PLOT 62,45:DRAWTO 70,49:DRAWTO 95,49:DRAWTO 70,49:DRAWTO 70,11

```

```

817 PLOT 62,15:DRAWTO 62,45:RETURN
818 PLOT 38,15:DRAWTO 30,11:DRAWTO 5,11:PLOT 38,45:DRAWTO 30,49:DRAWTO 5,49:DRAWTO 30,49:DRAWTO 30,11
819 PLOT 38,15:DRAWTO 38,45:RETURN
820 IF LOOKDIR=1 THEN LOCDR(1)=LOCW:LOCDR(2)=LOCN:LOCDR(3)=LOCE:LOCDR(4)=LOCS:LOCDR(5)=LOCNW:LOCDR(6)=LOCNE
826 IF LOOKDIR=2 THEN LOCDR(1)=LOCN:LOCDR(2)=LOCE:LOCDR(3)=LOCS:LOCDR(4)=LOCW:LOCDR(5)=LOCNE:LOCDR(6)=LOCSE
827 IF LOOKDIR=3 THEN LOCDR(1)=LOCE:LOCDR(2)=LOCS:LOCDR(3)=LOCW:LOCDR(4)=LOCN:LOCDR(5)=LOCSE:LOCDR(6)=LOCSW
828 IF LOOKDIR=4 THEN LOCDR(1)=LOCS:LOCDR(2)=LOCW:LOCDR(3)=LOCN:LOCDR(4)=LOCE:LOCDR(5)=LOCSW:LOCDR(6)=LOCNW
829 COLOR C:GOSUB 950
831 IF LOCDR(1)=1 THEN PLOT 5,1:DRAWTO 38,16:DRAWTO 38,44:DRAWTO 5,59
832 IF LOCDR(2)=1 THEN GOSUB 884
833 IF LOCDR(3)=1 THEN PLOT 95,1:DRAWTO 62,16:DRAWTO 62,44:DRAWTO 95,59
835 IF LOCDR(1)=0 AND LOCDR(5)=1 AND LOCDR(2)=0 THEN GOSUB 818
836 IF LOCDR(3)=0 AND LOCDR(6)=1 AND LOCDR(2)=0 THEN GOSUB 816
837 IF VIEWON=1 AND CH=1 THEN POKE DL+5,144:POKE DL+4,96
838 RETURN
840 POKE 89,144:POKE 88,96
842 COLOR C:PLOT 4,0:DRAWTO 96,0:DRAWTO 96,60:DRAWTO 4,60:DRAWTO 4,0
843 PLOT 118,10:DRAWTO 128,10:PLOT 123,15:DRAWTO 123,5
844 RETURN
870 IF VIEWON=1 THEN C=0:GOSUB 820
871 RETURN
884 PLOT 38,16:DRAWTO 62,16:PLOT 38,44:DRAWTO 62,44
885 IF LOCDR(1)=0 THEN PLOT 5,16:DRAWTO 40,16:PLOT 5,44:DRAWTO 40,44
886 IF LOCDR(3)=0 THEN PLOT 60,16:DRAWTO 95,16:PLOT 60,44:DRAWTO 95,44:RETURN
900 A=LOCY:IF A<1 THEN A=MAPY+LOCY:REM LOCN,LOCE,LOCS,LOCW COUNTING - BASED ON LOCX,LOCY
901 LOC=(A-2)*MAPX+LOCX-1
902 LOCNW=ASC(MAP$(LOC))-48:LOCN=ASC(MAP$(LOC+1))-48:LOCNE=ASC(MAP$(LOC+2))-48
903 LOC=LOC+MAPX:LOCW=ASC(MAP$(LOC))-48:LOCE=ASC(MAP$(LOC+2))-48
904 LOC=LOC+MAPX:LOCSW=ASC(MAP$(LOC))-48:LOCS=ASC(MAP$(LOC+1))-48:LOCSE=ASC(MAP$(LOC+2))-48
905 RETURN
906 PRINT :PRINT "Posiadasz:":GOSUB 6504:FOR I=1 TO MAXITEMS:INPUT #1,INV$
907 IF INV(I)>0 THEN PRINT INV$:INPUT VERB$
908 NEXT I:PRINT :CLOSE #1:RETURN
950 L=LOCY:IF L<0 THEN L=L+MAPY
951 I=0:IF LOOKDIR=1 AND L>2 AND LOCN=0 THEN I=(L-3)*MAPX+LOCX
952 IF (LOOKDIR=2 AND LOCE=0) OR (LOOKDIR=4 AND LOCW=0) THEN I=(L-1)*MAPX+LOCX+2:IF LOOKDIR=4 THEN I=I-4
953 IF LOOKDIR=3 AND L<MAPY-1 AND LOCS=0 THEN I=(L+1)*MAPX+LOCX
954 IF I>0 THEN IF MAP$(I,I)="1" THEN PLOT 39,20:DRAWTO 61,20:PLOT 39,40:DRAWTO 61,40
955 RETURN
1000 IF VIEWON=1 THEN COLOR 0:C=1:GOTO 1010
1001 IF MAPVISIBLE=1 THEN RETURN
1002 IF VIEWON=0 THEN COLOR 0:C=0
1010 PLOT 123,3:PLOT 130,10:PLOT 123,17:PLOT 116,10
1011 COLOR C
1012 IF LOOKDIR=1 THEN PLOT 123,3
1013 IF LOOKDIR=2 THEN PLOT 130,10
1014 IF LOOKDIR=3 THEN PLOT 123,17
1015 IF LOOKDIR=4 THEN PLOT 116,10
1020 RETURN
1200 CLOSE #1:OPEN #1,8,0,"D:FORT.SAV"
1201 A=LOCX:LOCX=MAPX(1):? #1;LOCX:LOCX=A:? #1;LOCX:? #1;LOCY
1202 FOR I=1 TO MAXITEMS:? #1;INV(I):? #1;LOCITEMX(I):? #1;LOCITEMY(I):? #1;ITEMINLOC(I):NEXT I
1203 ? #1;NOTE1$:? #1;SALA:? #1;DOOR:J=LOOKDIR(0):? #1;J:CLOSE #1:? "Zapisales stan gry. ":RETURN
1210 GOSUB 870:VIEWON=0:MAPVISIBLE=0:GOSUB 198:CLOSE #1:OPEN #1,4,0,"D:FORT.SAV"
1211 LOCX=MAPX(1):INPUT #1,LOCX:MAPX(1)=LOCX:INPUT #1,LOCX:INPUT #1,LOCY
1212 FOR I=1 TO MAXITEMS:INPUT #1,J:INV(I)=J:INPUT #1,J:LOCITEMX(I)=J:INPUT #1,J:LOCITEMY(I)=J:INPUT #1,J:ITEMINLOC(I)=J
1213 NEXT I:INPUT #1,NOTE1$:INPUT #1,SALA,DOOR,J:CLOSE #1:LOOKDIR(0)=J:? "Wczytywanie stanu gry. Czeka..."
1214 GOSUB 9010:GOSUB 6504:GOSUB 6505:GOSUB 6506:GOSUB 6507:GOSUB 6510:MAPNR=1
1215 IF LOCY>5 OR LOCY<0 THEN MAPNR=2
1216 ? :? "Stan gry wczytany poprawnie. Mozesz grac dalej...":GOSUB 810:?
1217 RETURN
6440 ? "Co bierziesz ";INPUT TMP$:IF TMP$="PLECAK" AND INV(9)=1 THEN GOSUB 270:GOTO 280

```


9029 ? "No graph view for the actual location. Fort Visit Permis d'Entrance Card. ";GOSUB 810
9030 RETURN

====

Skrócony opis kodu programu:

Linie 02 do 12: blok inicjalizacji stałych i zmiennych
Linie 49 do 72: strona tytułowa / informacyjna programu
Linie 80 do 99: ustawienie początkowych wartości stałych i zmiennych oraz inicjalizacja pierwszych funkcji programu
Linie 100 do 135: blok główny programu
Linie 198 do 9030: podprogramy i funkcje programu wywoływane z bloku głównego i z pozostałych bloków (wstępnych-inicjalizujących)
programu

DS.'.

MAR.08.2016.

Fort Quest 04-02 beta '2016 – trial tmp working version

Filename: FORT-QUEST-04-02B.atr

===